
Figure S1: NONMEM model code for gastric emptying (GE) and gallbladder emptying (GBE)

$PROBLEM Gastric emptying (GE) – Gallbladder emptying (GBE) model
$INPUT ID ; Subject identifier

STUDY ; Study flag (1: Study C, 2: Study B,
; 3: Study A, 4: Study D)

T2D ; Type 2 diabetic flag (0: healthy, 1: T2D)
DRINK ; Test drink flag (0: Water, 2: OGTT 75g,

; 4: Low fat, 5: Medium fat, 6: High fat)
TIM0 ; Time since last test drink intake (min)
TIME ; Time since beginning of the study

; (TIME = TIM0 - 20 min) (min)
DV ; Gallbladder volume (mL)
VAR ; Variable flag (1: dose, 3: gallbladder volume)
AMT ; Dose amount (i.e. calories of the test drink)
KCAL ; Caloric content of the test drink (kcal)
RATE ; Rate of dose input (0: bolus,

; -2: estimated duration)
EVID ; NONMEM event identifier (0: observation,

; 1: dose, 2: dummy rows, 4:dose w/ reset)
IT50 ; Individual values of the T50 parameter
IKG0 ; Individual values of the KG0 parameter
ISLPCAL ; Individual values of the SLPCAL parameter
IKUL ; Individual values of the KUL parameter
ISIG ; Individual values of the SIG parameter
SET50 ; Standard error of the T50 parameter
SEKG0 ; Standard error of the KG0 parameter
SESLPCAL ; Standard error of the SLPCAL parameter
SEKUL ; Standard error of the KUL parameter
SESIG ; Standard error of the SIG parameter
SEX ; Gender (1: Male, 2: Female)
AGE ; Age (yr)
WT ; Total body weight (kg)

$DATA data.csv
IGNORE=@ ; Ignore column headers
IGNORE=(STUDY.NE.1) ; Only keep data from study C

$SUBROUTINE ADVAN13 TOL=9

$MODEL COMP=(DEPOT) ; 1. Calories in stomach (kcal)
COMP=(UPPSI) ; 2. Calories in upper SI (kcal)
COMP=(DUOD) ; 3. Calories in duodenum (kcal)
COMP=(GBVOL) ; 4. Gallbladder volume (mL)

$PK
; Covariates ___________________________________________________________________
; BASE_B covariates
BASEWT = (1 + THETA(9)*(WT - 86)) ; Effect of WT on BASE_B (%/kg)

BASECOV = BASEWT

; S50_B covariates
S50AGE = (1 + THETA(10)*(AGE - 64)) ; Effect of AGE on S50_B (%/yr)
S50COV = S50AGE

; Gastric emptying _____________________________________________________________
; Fixed parameters
IF(STUDY.EQ.3) D1 = 1 ; Test drink infusion in study A (min)

; Population parameters with uncertainty (PPSE)
T50 = IT50 + ETA(1) * SET50 ; Time to 50% of maximal GE (min)
SIG = ISIG + ETA(2) * SESIG ; Sigmoidicity factor of the

; gastric emptying onset (_)
IF(T50.LE.0) T50 = 0
IF(SIG.LE.0) SIG = 0

; Individual parameters with uncertainty (IPPSE)
KG0 = IKG0 * EXP(ETA(3) * SEKG0) ; Baseline gastric emptying rate

; constant
SLPCAL = ISLPCAL * EXP(ETA(4) * SESLPCAL) ; Slope of the caloric feedback loop

; on gastric emptying
KUL = IKUL* EXP(ETA(5) * SEKUL) ; Nutrients transfer rate constant

; between upper and lower SI

; Calories parameters __________________________________________________________
; Saturable absorption of glucose from SI, values from Alskär et al. (2015)
; (conversion from glucose to kcal by using 4kcal/g of glucose)
RAMAX = 0.573 * 4 ; Maximal nutrient absorption rate (kcal)
KM = 6.28 * 4 ; Potency of caloric absorption (kcal)

F1 = 1 ; Bioavailability
IF(DRINK.EQ.0) F1 = 0 ; Get dose of 0 kcal for the water drink

; Gallbladder emptying parameters ______________________________________________
TVBASE_B = THETA(1) * BASECOV
BASE_B = TVBASE_B * EXP(ETA(6)) ; Baseline volume of gallbladder (mL)

TVKR_B = THETA(2)
KR_B = TVKR_B * EXP(ETA(7)) ; Rate constant of gallbladder

; emptying (min-1)

SMAX_B = THETA(3) ; Max calories signal effect on
; gallbladder emptying (_)

TVS50_B = THETA(8) * S50COV
S50_B = TVS50_B * EXP(ETA(8)) ; Signal that leads to 50% of max

; gallbladder emptying rate (_)

RPROD_B = BASE_B * KR_B ; Rate of bile production (mL/min)

A_0(4) = BASE_B ; Gallbladder baseline volume (mL)

KDJ = 1/12 ; Rate cst of transfert from duodenum
; to jejunum (min-1)

; Test drinks composition ______________________________________________________
IF(EVID.EQ.1.OR.EVID.EQ.4) TDOS = TIME ; Time of the dose (min)

; Amount of carbohydrates (g)
CARB = 0 ; Water test drink
IF(DRINK==2) CARB = 75 ; OGTT 75 g
IF(DRINK==4) CARB = 107 ; Low fat test drink
IF(DRINK==5) CARB = 93 ; Medium fat test drink
IF(DRINK==6) CARB = 32 ; High fat test drink

; Amount of proteins (g)
PROT = 0 ; OGTT 75 g / water test drink
IF(DRINK==4) PROT = 13 ; Low fat test drink
IF(DRINK==5) PROT = 11 ; Medium fat test drink
IF(DRINK==6) PROT = 3 ; High fat test drink

; Amount of fats (g)
FAT = 0 ; OGTT 75 g / water test drink
IF(DRINK==4) FAT = 2.5 ; Low fat test drink
IF(DRINK==5) FAT = 10 ; Medium fat test drink
IF(DRINK==6) FAT = 40 ; High fat test drink

; Nutrients potency
POTFAT = 1 ; Potency/g fats
POTPROT = THETA(4) ; Potency/g proteins
POTCARB = THETA(5) ; Potency/g carbohydrates

; Total nutrients effect on gallbladder emptying
EFAT = POTFAT * FAT ; Total effect of fats
EPROT = POTPROT * PROT ; Total effect of proteins
ECARB = POTCARB * CARB ; Total effect of carbohydrates

$DES
; Compartment amounts
X1 = A(1) ; Calories in stomach (kcal)
X2 = A(2) ; Calories in upper SI (kcal)
X3 = A(3) ; Calories in duodenum (kcal)
X4 = A(4) ; Gallbladder volume (mL)
IF(X2.LE.0) X2 = 0
IF(X3.LE.0) X3 = 0

3



X3S = 0 ; Signal of nutrients in duodenum (_)
IF(KCAL.GT.0) X3S = (X3/KCAL) * (EFAT + EPROT + ECARB)

; Onset of gastric emptying
TAD1 = T - TDOS ; Time after dose (min)
LAG1 = 1 ; Onset of gastric emptying
IF(STUDY.NE.3) THEN ; Not applied to water test drink
LAG1 = 1/(1+EXP(-SIG*(TAD1-T50)))

ENDIF

; Gastric emptying model _______________________________________________________
KG1 = KG0*(1 + SLPCAL*X2) ; Rate constant of GE (min-1)
RAD = RAMAX*X2/(KM+X2) ; Rate of calories absorption in

; upper SI (kcal/min)

DADT(1) = -KG1*X1*LAG1
DADT(2) = KG1*X1*LAG1 -RAD -KUL*X2 ; Calories in upper SI (kcal)

; Gallbladder emptying Model ___________________________________________________
; Differential Eq
B_REL = 1 + (SMAX_B*X3S)/(S50_B+X3S) ; Effect of nutrients in duodenum

; on gallbladder emptying
RAD2 = RAMAX*X3/(KM+X3) ; Rate of calories absorption in

; duodenum (kcal/min)

DADT(3) = KG1*X1*LAG1 -RAD2 -KDJ*X3 ; Calories in duodenum (kcal)
DADT(4) = RPRO_B - X4*KR_B*B_REL ; Gallbladder volume (mL)

$ERROR
; Compartment amounts
A1 = A(1) ; Calories in stomach (kcal)
A2 = A(2) ; Calories in upper SI (kcal)
A3 = A(3) ; Calories in duodenum (kcal)
A4 = A(4) ; Gallbladder volume (mL)
IF(A2.LE.0) A2 = 0
IF(A3.LE.0) A3 = 0

A3S = 0 ; Signal of nutrients in duodenum (_)
IF(KCAL.GT.0) A3S = (A3/KCAL)*(EFAT + EPROT + ECARB)

; Onset of gastric emptying
TAD = TIME - TDOS ; Time after dose (min)
LAG = 1 ; Onset of gastric emptying
IF(STUDY.NE.3) THEN ; Not applied to water test drink
LAG = 1/(1+EXP(-SIG*(TAD-T50)))

ENDIF

4



IPRED = A4 ; Individual predictions
IRES = DV - IPRED ; Individual residuals
W = SQRT((THETA(6)*IPRED)**2 + THETA(7)**2) ; Residuals weight
IWRES = IRES/W ; Individual weighted residuals
Y = IPRED + W*EPS(1) ; Gallbladder volume fit

$THETA
(0,36.3098) ; Th1. BASE (mL)
(0,0.0688355) ; Th2. KR_B (min-1)
(0,6.47407) ; Th3. SMAX_B (_)
(0,0.685419) ; Th4. Relative potency proteins (g-1)
(0,0.0222451) ; Th5. Relative potency carbohydrates (g-1)
(0,0.076712) ; Th6. Proportional Error (%)
(0,2.33466) ; Th7. Additive Error (mL)
(0,5.49109) ; Th8. S50_B (_)
0.0118959 ; Th9. WT on BASE_B (%/kg)
0.0231464 ; Th10. Age on S50_B (%/yr)

$OMEGA
1 FIX ; Om1.1. SE_T50
1 FIX ; Om2.2. SE_SIG
1 FIX ; Om3.3. SE_KG0
1 FIX ; Om4.4. SE_SLPCAL
1 FIX ; Om5.5. SE_KUL
0.0725307 ; Om6.6. BASE_B
0.715303 ; Om7.7. KR_B
0.660614 ; Om8.8. S50_B

$SIGMA
1 FIX ; Sig1.1. Residual error

$ESTIMATION METHOD=1 SIGL=9 NSIG=3 INTER MAXEVAL=9999 PRINT=1 NOABORT
$ESTIMATION METHOD=IMP EONLY=1 ISAMPLE=1000 NITER=5

$COVARIANCE UNCONDITIONAL

$TABLE ID TIME AMT EVID IPRED CWRES IWRES A1 A2 A3 A4 A3S ETAS(1:LAST)
NOPRINT ONEHEADER FILE=sdtab102

$TABLE ID KG0 T50 SLPCAL KUL SIG BASE_B KR_B SMAX_B S50_B
NOAPPEND NOPRINT ONEHEADER FILE=patab102

$TABLE ID STUDY T2D DRINK KCAL SEX
NOAPPEND NOPRINT ONEHEADER FILE=catab102

$TABLE ID AGE WT
NOAPPEND NOPRINT ONEHEADER FILE=cotab102

